2.3.2

ICT ENABLED TEACHING SUPPORTIVE DOCUMENT

LAB MANUAL

DEPT OF MATHEMATICS (UG)

08256 - 236220, 236101 (O) 236 220 sdmcollege@sdmcujire.in www.sdmcujire.in

Lab Manual for
III Semester B. Sc. Mathematics Practical
Mangalore University

Price : Rs . 20

Lab Manual for
III Semester B. Sc. Mathematics Practical
Mangalore University

Price : Rs . 20

Index

I (1)Solutions of ordinary differential equations of first and second order.

An ordinary differential equation is an equation involving derivatives of a function of a single variable. A solution is either an explicit function satisfying the equation, or an implicit relation between the independent and dependent variables.

A general solution of an ODE involves some arbitrary constants. If some initial conditions like $y(x_0) = y_0$ and $y'(x_0) = dy_0$, or some boundary conditions like $y(x_0) = y_0$ and $y(x_1) = y_1$ are given, then by substituting these in the general solution we get the particular solution with out the arbitrary constant.

Maxima commands used: Function **ode2( )**, variables **method**, **intfactor**, **odeindex**, **yp**, functions **declare ( )**, **ic1( )**, **bc2( )**, **contrib_ode( )**

Function: **ode2 (eqn, dvar, ivar)**

The function **ode2 ( )** solves an elementary ordinary linear differential equation (ODE) of first or second order, and gives the general solution. It takes three arguments: an ODE given by **eqn**, the dependent variable **dvar**, and the independent variable **ivar**. When successful, it returns either an explicit or implicit solution. %c is used to represent the integration constant in the case of first-order equations, and %k1 and %k2 the constants for second-order equations.

eqn should contain only derivatives got by **diff ()** and not differentials like dx and dy.

E.g. 1. To solve the differential equation $x^2\frac{dy}{dx} + 3xy = \frac{\sin x}{x}$

The derivative $\frac{dy}{dx}$ has to be written as **'diff(y, x)** , and not as **diff(y,x)**. It is because y is not given as an explicit function of x, and so **diff(y,x)** evaluates to 0. To suppress the evaluation, the apostrophe has to be used.

x^2*'diff(y,x) + 3*y*x = sin(x)/x;

$$(\%o1)\quad x^2\left(\frac{d}{d\,x}\,y\right)+3\,x\,y=\frac{\sin(x)}{x}$$

Now to solve it

ode2(%, y, x); Here **%** indicates the previous output, that is, the given equation. The result is

$$(\%o2)\quad y=\frac{\%c-\cos(x)}{x^3}$$

Instead of using **%** to take the previous output, we can also pass the equation directly to the function **ode2( )** , as follows.

kill(all)$
ode2 (x^2*'diff(y,x) + 3*y*x = sin(x)/x, y,x);

$$(\%o1)\quad y=\frac{\%c-\cos(x)}{x^3}$$

Or, we can assign the equation to a variable say, **eqn**, and pass **eqn** to the function **ode2( )**, as follows.

kill(all)$
eqn: x^2*'diff(y,x) + 3*y*x = sin(x)/x$
ode2(eqn, y,x);

(%o2) $y=\frac{\%c-\cos(x)}{x^3}$

If **ode2( )** cannot obtain a solution for whatever reason, it returns false, after perhaps printing out an error message.

E.g. 2: To solve $x^2\frac{dy}{dx}+3xy=\frac{\sin x}{x}$

kill(all)$
x^2*diff(y,x) +3*y*x=sin(x)/x;
ode2(%,y,x);

(%o1) $3\,x\,y=\frac{\sin(x)}{x}$

(%t2) $3\,x\,y=\frac{\sin(x)}{x}$

not a proper differential equation
(%o2) ***false***

For first order ordinary differential equations, **ode2()** implements the following methods in the following order : linear, separable, exact - perhaps requiring an integrating factor, homogeneous, Bernoulli's equation, and a generalized homogeneous method.

Some types of second-order equations which can be solved are: constant coefficients, linear homogeneous with non-constant coefficients which can be transformed to constant coefficients, equations solvable by the method of variation of parameters, and equations which are free of either the independent or of the dependent variable so that they can be reduced to two first order linear equations to be solved sequentially.

In the course of solving ODE's, several variables are set purely for informational purposes:

method denotes the method used (e.g., **linear, variationofparameters** etc.)

intfactor denotes any integrating factor used,

odeindex denotes the index for Bernoulli's method or for the generalized homogeneous method

yp denotes the particular solution for the variation of parameters technique.

By printing the values of the above variables after solving the ODE, we can check which method was used.

E.g. 3. Solve $\frac{dy}{dx} + \left(\frac{1-y^2}{1-x^2}\right)^{\frac{1}{2}} = 0$ using Maxima, and find the method used.

```
kill(all)$
eqn: 'diff(y,x) + sqrt(  (1-y^2))/  sqrt((1-x^2) )=0;
s:ode2( eqn, y,x);
method;
```

(eqn) $\frac{d}{d\,x}\,y+\frac{\sqrt{1-y^2}}{\sqrt{1-x^2}}=0$

(s) $-\operatorname{asin}(y)=\operatorname{asin}(x)+\%c$

(%o3) *separable*

(Note: If we use sqrt((1-y^2)/(1-x^2)) in the equation, we do not get a proper solution. In expressions, as much as possible keep the numerator and denominator separate.)

Generally if we use any names like $a, b, theta$ etc., They are considered as variables. But sometimes we need such names as constants in an ODE.

declare (a1, p1, a2, p2 ,...)

assigns the property **pi** to the atom **ai** for each **i**. (An atom is a number, name or string.) When **ai** is a number, some of the properties that can be used are **constant, integer, even, odd, real** etc.

E.g. 4. To solve $\frac{dy}{dx} + a\cos x = 1$

```
declare( a , constant)$
eqn: 'diff(y,x)+a* cos(x)=1 ;
soln:ode2( eqn, y,x);
```

(eqn) $\frac{d}{dx}y + a\cos(x) = 1$

(soln) $y = -a\sin(x) + x + \%c$

ic1 (solution, x=x0, y=y0) Solves initial value problems for first order differential equations. Here **solution** is a general solution of a differential equation as found by **ode2( )**, and the initial condition is $y(x0) = y0$.

E.g. 5. Solve $\frac{dy}{dx} + \left(\frac{1-y^2}{1-x^2}\right)^{\frac{1}{2}} = 0\,,\, y(0) = \frac{1}{2}$. Mention the method used.

```
kill(all)$
eqn: 'diff(y,x)+sqrt(  (1-y^2))/  sqrt((1-x^2) )=0;
s1:ode2( eqn, y,x);
s2:ic1(s1, x=0, y=1/2);
method;
```

(eqn) $\frac{d}{dx}y + \frac{\sqrt{1-y^2}}{\sqrt{1-x^2}} = 0$

(s1) $-\mathrm{asin}(y) = \mathrm{asin}(x) + \%c$

(s2) $-\mathrm{asin}(y) = \frac{6\,\mathrm{asin}(x) - \pi}{6}$

(%o4) *separable*

ic2 (solution, x=x0, y=y0, `diff(y,x) =dy0) solves initial value problems for second-order differential equations. Here **solution** is a general solution to the equation, as found by **ode2()**, and the initial condition is $y(x0) = y0,\ y'(x0) = dy0$.

E.g. 6. Solve $\frac{d^2y}{dx^2} - x^2 = 0$, with initial conditions $y(1) = 3$, $y'(1) = 2$. Give also the general solution and print the method used.

```
kill(all) $
eqn: 'diff(y,x,2)-x^2=0;
s: ode2(eqn, y,x);
method;
ic2(s, x=1,y=3, 'diff(y,x) =2);
```

(eqn) $\frac{d^2}{dx^2} y - x^2 = 0$

(s) $y = \frac{x^4}{12} + \%k2\, x + \%k1$

(%o3) *variationofparameters*

(%o4) $y = \frac{x^4}{12} + \frac{5x}{3} + \frac{5}{4}$

bc2 (solution, x=x1, y=y1, x=x2, y=y2)

Solves a boundary value problem for a second order differential equation. Here **solution** is a general solution to the equation, as found by **ode2( )**, and the boundary conditions are $y(x1) = y1, y(x2) = y2$.

E.g. 7. Solve $\frac{d^2y}{dx^2} - x^2 = 0$, with boundary conditions $y(0) = 3$, $y(1) = 4$. Give also the general solution and print the method used.

```
kill(all) $
```

```
eqn: 'diff(y,x,2)-x^2=0;
s: ode2(eqn, y,x);
method;
bc2(s, x=0,y=3, x=1,y=4);
```

(eqn) $\frac{d^2}{d x^2} y - x^2 = 0$

(s) $y = \frac{x^4}{12} + \%k2\, x + \%k1$

(%o3) *variationofparameters*

(%o4) $y = \frac{x^4}{12} + \frac{11\, x}{12} + 3$

Since we now know that the method used is variation of parameters, we can also find the particular solution used.

yp;

(%o5) $\frac{x^4}{12}$

contrib_ode( ): This can be used to solve some ODEs that cannot be solved by **ode2( )**, i.e. non-linear first order ODEs and linear homogeneous second order ODEs. For this we have to load the package **contrib_ode**, by

load('contrib_ode)$ (or **load("contrib_ode"**)

If **eqn** is an ODE with independent variable *x* and dependent variable *y*, **contrib_ode(eqn, y, x)** returns a list of solutions of **eqn** .

E.g. 8. Find the general solution $y' = \frac{1-xy^2}{2x^2y}$.

```
load('contrib_ode)$
eqn: 'diff(y,x)=(1-x*y^2)/(2*x^2*y);
s1: contrib_ode(eqn, y,x);
```

(eqn) $\frac{d}{dx}y=\frac{1-xy^2}{2x^2y}$

(s1) $[xy^2-\log(x)=\%c]$

Note that **contrib_ode( )** gives an array of one or more solutions.

<u>Practical 1)</u> <u>General and particular solutions of ordinary differential equations:</u> Rewrite the following ordinary differential equation as required by Maxima, find all the general solutions and mention the method used. Also find the particular solution with the given initial/boundary conditions : $\left(\frac{dy}{dx}\right)^2+x\frac{dy}{dx}-y=0$, with initial condition $y(0)=1$.

```
kill(all)$
load('contrib_ode)$
eqn:'diff(y, x)^2+x*'diff(y, x)-y=0;
s1:ode2( eqn, y, x);
```

(eqn) $\left(\frac{d}{dx}y\right)^2+x\left(\frac{d}{dx}y\right)-y=0$

(s1) $\left(\frac{d}{dx}y\right)^2+x\left(\frac{d}{dx}y\right)-y=0$

first order equation not linear in y'

(s1) ***false***

/* ode2( ) does not work, this part may be omitted */

```
s1:contrib_ode( eqn, y, x);
```

(s1) $\left(\frac{d}{dx}y\right)^2+x\left(\frac{d}{dx}y\right)-y=0$

first order equation not linear in y'

(s1) $[y=\%c\,x+\%c^2, y=-\frac{x^2}{4}]$

method;

(%o5) *clairault*

s1[1];

(%o6) $y=\%c\,x+\%c^2$

/* first general solution, with arbitrary constant */

s_ip: ic1(s1[1], x=0,y=1);

(s_ip) $y=1-x$

/* particular solution of initial value problem */

s1[2];

(%o8) $y=-\frac{x^2}{4}$ /* second solution, without arbitrary constants */

Exercise 1

Rewrite the following ordinary differential equations as required by Maxima, find all the general solutions, print the method used. Also find the particular solution with the given initial/boundary conditions.

1) $(x+1)dy=(2e^y-1)dx$, with initial condition $y(2)=3$.

2) $\frac{dy}{dx}+\frac{10x+8y-12}{7x+5y-9}=0$, with initial condition $y(0)=5$.

3) $\frac{dy}{dx} + y\tan x = \cos^3 x$, with $y=0$ when $x = \frac{\pi}{4}$.

4) $(1+x^2)dy = \left(e^{\tan^{-1}x} - y\right)dx$, with $y(0)=4$.

5) $(D^2 - 5D + 4)y = 0$, with $y(0) = 4$, $y'(0) = 1$.

6) $dx = t(1+t^2)\sec^2 x\, dt$, $y(1)=1$.

7) $xyp^2 + (x+y)p + 1 = 0$, with $y(4) = -5$, where p is $\frac{dy}{dx}$.

8) $x(y')^2 + 4y' - 6x^2 = 0$, with $y(1) = 0$.

9) $2y'' + 4y' = e^x$, with $y(0) = \frac{1}{6}$ and $y(1) = \frac{e}{6}$.

10)) $(x+8)dy = (6e^{-y} - 1)dx$, with initial condition $y(1) = 5$.

LAB MANUAL
For Second Semester
B.Sc. Mathematics Practical

As Prescribed by
Mangalore University

Name :

Reg. No. :

College :

Prepared by
Forum of Mathematics Teachers (FORMAT)

Lab Manual for
II Semester B. Sc. Mathematics Practical
Mangalore University

Price: Rs. 20

Index

I Limits

I (1) Limits, continuity and differentiability at a finite number:

A function is continuous at a finite number 'a' if both the left hand side and right hand side limits of $f(x)$ exist at a, and both are equal to $f(a)$. The function is differentiable at a if both L.H. and R.H. limits of $\frac{f(x)-f(a)}{x-a}$ exist and are equal.

Maxima commands needed: **limit(), set( ), elementp( ).**

We use the concept of sets in maxima. We use set inclusion in order to avoid too many if – then statements.

set(a_1, ..., a_n) constructs a set with members a_1, ..., a_n,
We can get the same result by writing **{a_1, ..., a_n};**

To construct the empty set, write **set()** or {}. For input, set (...) and {... } are equivalent. For output, sets are always displayed with curly braces.

If a member is listed more than once, the redundant member is automatically eliminated.

E.g. 1.

```
kill(all)$
set (a,b,c);
(%o1) {a,b,c}
{a,b,c};
(%o2) {a,b,c}
```

E.g. 2.

s1: set(x,y,z);

s2: set();

s3:{};

```
(s1)    {x,y,z}
(s2)    {}
(s3)    {}
```

E.g. 3.

s: set(a, b, c, a);

```
(s)    {a,b,c}
```

Here note that '*a*' is written only once.

elementp(x, s) returns **true** if and only if ***x*** is a member of the set **s.** elementp ( ) complains if **s** is not a set.

E.g. 4.

s: set(a,b,c,d);

elementp(a,s); elementp(x,s);

```
(s)    {a,b,c,d}
(%o2)  true
(%o3)  false
```

We assume that when '*a*' is a finite number, if a limit as *x* tends to '*a*' exists and equals $+\infty$ or $-\infty$, or if it is indefinite and unbounded, or if it is indefinite and bounded, we should get the output as "limit does not exist".

When '*a*' is $+\infty$ or $-\infty$, we can check the one sided limit. We assume that if the limit is finite, $+\infty$ or $-\infty$, we should get the output as " limit exists". If it is bounded and indefinite or unbounded and indefinite , the output should be "limit does not exist".

In both cases when the function is not defined at *a*, and it is defined at *a* but is discontinuous at *a*, the output should be " f is discontinuous at a".

Maxima computes limits of indeterminate forms using L-Hospital's rule.

To check whether a function $f(x)$ is continuous at a point $x = a$, first we check whether both LH and RH limits of $f(x)$ at 'a' are defined and finite. This will be true if the limits do not belong to the set s= { **inf, minf, infinity, und, ind**}. If both limits are not in this set, then we can check for continuity and differentiability as usual.
We use a variable 'flag', which is set to 1 initially, assuming that the function is continuous. In case any of the one-sided limits are in the set s, or if the limits are different, then 'flag' is made 0, indicating that the function is discontinuous. Only if flag is 1, differentiability is to be checked in the same way, by finding limits of $\frac{f(x)-f(a)}{x-a}$.

Practical 1: Check the continuity of $f(x) = \begin{cases} \frac{x}{\sin x}, & x < 0 \\ 1 + 3x, & x \geq 0 \end{cases}$ at x=0.

If continuous, check for differentiability.

(First write the code using ; to check intermediate results. Then they can be replaced by $)

```
kill(all)$
s:{ inf, minf,  infinity , und, ind }$
flag:1$ a:0$
 f(x):=  if x<a then x/sin(x) else 1+3*x $
 f1(x) := x/sin(x) ; f2(x) :=3*x+1;
 l1: limit ( f1(x), x,a, minus) $  l2: limit ( f2(x), x,a,plus)$
   if elementp(l1,s)  then
   (flag:0, print ( " LH limit does not exist ,
                                    function is not continuous at  ", x=a))$

 if elementp(l2,s) then
```

```
(flag:0,   print ( " RH limit does not exist ,
                        function is not continuous at  ", x=a))$


if flag=1 then
(     print ("LH limit is ", l1, "  RH limit is   ", l2),

      if l1=l2 then
      (      print ( " Function is continuous at   ", x=a ,
                          " function value is  ", l1 ),
             dl1: limit ( (f1(x) -l1)/(x-a),x, a, minus ),
             dl2: limit ( (f2(x) -l1)/(x-a), x, a,  plus  ),
             if  elementp(dl1,s) or elementp(dl2,s) or dl1#dl2 then
                    print ( "  Function is not differentiable at  ", x=a)
             else        print ( " Function is differentiable at  ", x=a,
                                                          "   f ' is =  ", dl1)

      )
      else print (" Function is not continuous at ", x=a)
)$
```

(A draw statement can be included at the end if needed, to demonstrate the graph)

```
wxdraw2d( xrange= [a-5,a+5], yrange=[ a-5,a+5] ,
      explicit (f1(x),  x, a-5, a) ,explicit( f2(x), x, a, a+5)) $
```

(%o5) $f1(x):=\frac{x}{\sin(x)}$

(%o6) $f2(x):=3x+1$

LH limit is 1 *RH limit is* 1

Function is continuous at $x=0$ *function value is* 1

Function is not differentiable at $x=0$

I(2) Limits at infinity

Practical 2: Find the limits of $f(x) = \sin x + e^x$ as x tends to $+\infty$ and $-\infty$ if they exist.

```
kill(all)$
f(x):= sin(x)+exp(x);
a:inf$ b:minf$
s: {und,ind}$
 l1: limit(f(x), x,a)$
l2:limit ( f(x), x, b)$
if elementp(l2,s) then
   print( "limit of f( x) as x  -> ", b ,"does not exist" )
else
   print( "limit of f( x) as  x -> ", b ,  "is", l2)$
if elementp(l1,s) then
   print( "limit of f( x) as  x -> ", a ,"does not exist" )
else
   print( "limit of f( x) as  x -> ", a ,  "is", l1)$
```

```
(%o1)  f(x):=sin(x)+exp(x)
limit of f( x) as x →   -∞ does not exist
limit of f( x) as x→    ∞ is ∞
```

Ex. 1 :

Check the continuity of the following functions at the given points using Maxima. If continuous, check for differentiability at those points.

1) $f(x) = \begin{cases} x^2, & x \le 0 \\ \frac{1}{x^3}, & x > 0 \end{cases}$ at $x = 0.$

2) $f(x) = \sin\frac{1}{x}$ at $x = 0.$

3) $f(x) = \begin{cases} \frac{x-1}{\cos\frac{\pi x}{2}}, & x < 1 \\ -x-1, & x \geq 1 \end{cases}$ at $x = 1$.

4) $f(x) = \begin{cases} 2x-3, & x \leq 2 \\ \frac{\log(x-1)}{x-2}, & x > 2 \end{cases}$ at $x = 2$.

5) $f(x) = \sqrt[3]{x}$ at $x=0$.

6) $f(x) = \frac{1}{x^2-4}$ at $x=2$ and -2

7) $f(x) = \frac{\sec x}{1+\tan x}$ at $x = \frac{\pi}{2}$.

8) $f(x) = \frac{2x^2-(3x+1)\sqrt{x}+2}{x-1}$ at $x = 1$.

Find the limits of the following functions,

9) $f(x) = \sin x$ as $x \to +\infty$ and $-\infty$.

10) $f(x) = \frac{e^x}{x^n}$ for $n = 3, 5, 10, 20$ as $x \to +\infty$.

11) $f(x) = x^n \log x$ for $n = 2, 5, 10$ as $x \to +\infty$.

12) $f(x) = \frac{x-2x^2}{3x^2+5x}$ as $x \to +\infty$ and $-\infty$.

13) $f(x) = \frac{5x^2-3x}{7x^2+1}$ as $x \to +\infty$ and $-\infty$.